	*****************************************************************************************************
	* Inter-regional Inequality and the Dynamics of Government Spending 
	* Online Appendix -- Section 3. Alternative Measure of Regional Economy: Regional Unemployment Risk
	*****************************************************************************************************
	clear all
	set more off 
	set mem 600m
	cd "YOUR_DIRECTORY_PATH_HERE\supplements (online appendix)\appendix e"
	use "onlineappendix.dta"   

 	****************************************************
	** Generating / Rescaling / Modifying Variables                             
	****************************************************
	sort id year 
	sum ideal_pt_V1, meanonly
	gen jacoby_modified2 = ideal_pt_V1 - `r(mean)'
	replace adgini = adgini*100
	replace jacoby_modified2 = jacoby_modified2*100
	replace ineq_spns = ineq_spns*100
	gen log_pop = log(pop)	
	gen log_rgdpch = log(rgdpch)
	gen pop14_65 = popunder14 + popover65
    replace cov_unemp = cov_unemp*100
	replace covw_unemp = covw_unemp*100
	
	************************************
	* Generate five-year moving average 
	************************************
	* Generate averages for redistribution models
	foreach x of varlist adgini rpop_s1_rgdppc log_pop pop14_65 a log_rgdpch gov_left2 cov_unemp covw_unemp {
	gen ma`x' = (l1.`x' + l2.`x' + l3.`x' + l4.`x'+ l5.`x')/5
	}
	
	**************************
	** Renaming Variables 
	**************************
	rename adgini rdgini   /*For Footnote #5 in Appendix E*/
	rename malog_pop population_logged
	rename mapop14_65 dependent_population
	rename maa economic_globalization 
	rename malog_rgdpch ppp_gdppc_logged
	rename magov_left2 leftistgovt
	rename Parl parliamentary
	rename Unit2 nonfed_nonbicam
	rename govexp_central_cofog2 central_govts
	rename jacoby_modified2 policy_priority
	rename macov_unemp cov_unemp2
	rename macovw_unemp covw_unemp2
	
	* Spearman's Correlation 
	spearman cov_unemp covw_unemp rdgini if year>=1991 & year<=2011, star(.05) pw bonferroni stats(rho obs p)

	********************************************************
	** Appendix E. Regional Variation in Unemployment Risk
	********************************************************
	tab countries, gen(kcountry_)	
	global baseline l.population_logged l.d.population_logged l.dependent_population l.d.dependent_population l.economic_globalization l.d.economic_globalization l.ppp_gdppc_logged l.d.ppp_gdppc_logged l.leftistgovt

	** Models for Change in Central Government Spending
	* Model [1]
	xi:xtpcse d.central_govts l.central_govts l.cov_unemp2 $baseline l.PR l.parliamentary l.nonfed_nonbicam kcountry_* if year>=1991 & year<=2011,noconst pairwise corr(psar1)
	
	* Model [2]
	xi:xtpcse d.central_govts l.central_govts l.covw_unemp2 $baseline l.PR l.parliamentary l.nonfed_nonbicam kcountry_* if year>=1991 & year<=2011,noconst pairwise corr(psar1)
	
	** Models for Change in Policy Priority 
	* Model [3]
	xi:xtpcse d.policy_priority l.policy_priority l.cov_unemp2 $baseline l.PR l.parliamentary l.nonfed_nonbicam kcountry_* if year>=1991 & year<=2011,noconst pairwise corr(psar1)
	
	* Model [4]
	xi:xtpcse d.policy_priority l.policy_priority l.covw_unemp2 $baseline l.PR l.parliamentary l.nonfed_nonbicam kcountry_* if year>=1991 & year<=2011,noconst pairwise corr(psar1)

